Computations supporting Example 3.5 in the paper Matricial Gaussian Quadrature Rules: Nonsingular Case by Aljaž Zalar and Igor Zobovič.

This example shows a sequence S with k > 0, where k = rank (HH K) - rank HH, where HH is the (x - t)-localizing matrix. It exhibits two distinct ((n + 1)p)-atomic R-representing matrix measures for S. In both cases, the measures include an atom at t with the largest multiplicity allowed by Theorem 3.1, namely m = p - k, demonstrating that a representing measure for S that contains an atom t of multiplicity m is not unique if m < p.
Let p = 2, n = 1 and t = 0. Let S = (S0, S1, S2), where S0 := {{18,10},{10,7}}, S1 := {{2,2},{2,2}}, S2 := {{50,26},{26,14}}. Let M :=M(1) the corresponding moment matrix.

In[∘]:=
```
S0 = {{18, 10}, {10, 7}};
MatrixForm[S0]
S1 = {{2, 2}, {2, 2}};
MatrixForm[S1]
S2 = {{50, 26}, {26, 14}};
MatrixForm[S2]
M = ArrayFlatten[{{S0, S1}, {S1, S2}}];
MatrixForm[M]
```

Out[∘]//MatrixForm=

$$\begin{pmatrix} 18 & 10 \\ 10 & 7 \end{pmatrix}$$

Out[∘]//MatrixForm=

$$\begin{pmatrix} 2 & 2 \\ 2 & 2 \end{pmatrix}$$

Out[∘]//MatrixForm=

$$\begin{pmatrix} 50 & 26 \\ 26 & 14 \end{pmatrix}$$

Out[∘]//MatrixForm=

$$\begin{pmatrix} 18 & 10 & 2 & 2 \\ 10 & 7 & 2 & 2 \\ 2 & 2 & 50 & 26 \\ 2 & 2 & 26 & 14 \end{pmatrix}$$

We check that M is positive definite by computing its eigenvalues.

In[∘]:=
```
N[Eigenvalues[M]]
```

Out[∘]=
```
{63.9786, 23.6284, 1.17717, 0.215786}
```

Let HH = S1 be the (x - t)-localizing matrix and let K = S2. We will compute k := rank(HH K) - rank (HH).

In[]:=
```
HH = S1;
K = S2;
HHK = ArrayFlatten[{{HH, K}}];
MatrixForm[HHK]
MatrixRank[HHK]
MatrixRank[HH]
k = MatrixRank[HHK] - MatrixRank[HH]
```

Out[]//MatrixForm=

$$\begin{pmatrix} 2 & 2 & 50 & 26 \\ 2 & 2 & 26 & 14 \end{pmatrix}$$

Out[]=

2

Out[]=

1

Out[]=

1

We write K = (K1 K2). We check that rank(HH K) = rank(HH K1).

In[]:=
```
K1 = K[[All, 1 ;; k]];
MatrixForm[K1]
K2 = K[[All, k + 1 ;;]];
MatrixForm[K2]
HHK1 = ArrayFlatten[{{HH, K1}}];
MatrixForm[HHK1]
MatrixRank[HHK1]
```

Out[]//MatrixForm=

$$\begin{pmatrix} 50 \\ 26 \end{pmatrix}$$

Out[]//MatrixForm=

$$\begin{pmatrix} 26 \\ 14 \end{pmatrix}$$

Out[]//MatrixForm=

$$\begin{pmatrix} 2 & 2 & 50 \\ 2 & 2 & 26 \end{pmatrix}$$

Out[]=

2

We define J := {{-1/2},{1},{1/2}}, and we check that K2 = (HH K1)J.

In[]:=
```
J = {{-1 / 2}, {1}, {1 / 2}};
MatrixForm[J]
MatrixForm[HHK1.J]
```

Out[]//MatrixForm=

$$\begin{pmatrix} -\frac{1}{2} \\ 1 \\ \frac{1}{2} \end{pmatrix}$$

Out[]//MatrixForm=

$$\begin{pmatrix} 26 \\ 14 \end{pmatrix}$$

We will now construct two different matrices Z1 and Z2, which will represent S3.

In[]:=
```
Z11 = 2
Z12 = ArrayFlatten[{{Transpose[K1], Z11}}].J; MatrixForm[Z12]
Z13 = ArrayFlatten[{{Transpose[K2], Z12}}].J; MatrixForm[Z13]
Z1 = ArrayFlatten[{{Z11, Z12}, {Transpose[Z12], Z13}}]; MatrixForm[Z1]
Z21 = 98
Z22 = ArrayFlatten[{{Transpose[K1], Z21}}].J; MatrixForm[Z22]
Z23 = ArrayFlatten[{{Transpose[K2], Z22}}].J; MatrixForm[Z23]
Z2 = ArrayFlatten[{{Z21, Z22}, {Transpose[Z22], Z23}}];
MatrixForm[Z2]
```

Out[]=

2

Out[]//MatrixForm=

$$\begin{pmatrix} 2 \end{pmatrix}$$

Out[]//MatrixForm=

$$\begin{pmatrix} 2 \end{pmatrix}$$

Out[]//MatrixForm=

$$\begin{pmatrix} 2 & 2 \\ 2 & 2 \end{pmatrix}$$

Out[]=

98

Out[]//MatrixForm=

$$\begin{pmatrix} 50 \end{pmatrix}$$

Out[]//MatrixForm=

$$\begin{pmatrix} 26 \end{pmatrix}$$

Out[]//MatrixForm=

$$\begin{pmatrix} 98 & 50 \\ 50 & 26 \end{pmatrix}$$

We will now obtain two pairs of matrix coefficients (H10, H11) and (H20, H11) which will compose the corresponding generating polynomials.

```
In[]:= H1 = Inverse[M].ArrayFlatten[{{S2}, {Z1}}]; MatrixForm[H1]
H10 = H1[[1 ;; 2]]; MatrixForm[H10]
H11 = H1[[3 ;; 4]]; MatrixForm[H11]
polH1[x_] := x^2 IdentityMatrix[2] - (H10 + x H11)
polH1[x_] // Simplify // MatrixForm
H2 = Inverse[M].ArrayFlatten[{{S2}, {Z2}}]; MatrixForm[H2]
H20 = H2[[1 ;; 2]]; MatrixForm[H20]
H21 = H2[[3 ;; 4]]; MatrixForm[H21]
polH2[x_] := x^2 IdentityMatrix[2] - (H20 + x H21)
polH2[x_] // Simplify // MatrixForm
```

Out[]//MatrixForm=

$$\begin{pmatrix} 3 & \frac{3}{2} \\ 0 & 0 \\ 2 & \frac{1}{2} \\ -4 & -1 \end{pmatrix}$$

Out[]//MatrixForm=

$$\begin{pmatrix} 3 & \frac{3}{2} \\ 0 & 0 \end{pmatrix}$$

Out[]//MatrixForm=

$$\begin{pmatrix} 2 & \frac{1}{2} \\ -4 & -1 \end{pmatrix}$$

Out[]//MatrixForm=

$$\begin{pmatrix} -3 - 2\,x\_ + x\_^2 & \frac{1}{2}\,(-3 - x\_) \\ 4\,x\_ & x\_\,(1 + x\_) \end{pmatrix}$$

Out[]//MatrixForm=

$$\begin{pmatrix} 3 & \frac{3}{2} \\ 0 & 0 \\ 6 & \frac{5}{2} \\ -8 & -3 \end{pmatrix}$$

Out[]//MatrixForm=

$$\begin{pmatrix} 3 & \frac{3}{2} \\ 0 & 0 \end{pmatrix}$$

Out[]//MatrixForm=

$$\begin{pmatrix} 6 & \frac{5}{2} \\ -8 & -3 \end{pmatrix}$$

Out[]//MatrixForm=

$$\begin{pmatrix} -3 - 6\,x\_ + x\_^2 & \frac{1}{2}\,(-3 - 5\,x\_) \\ 8\,x\_ & x\_\,(3 + x\_) \end{pmatrix}$$

We compute the determinants from which we get the atoms.

In[∘]:=
```
Factor[Det[polH1[x]]]
Solve[Det[polH1[x]] == 0, x]
Factor[Det[polH2[x]]]
Solve[Det[polH2[x]] == 0, x]
```

Out[∘]=

$(-1 + x)\, x\, (-3 + x^2)$

Out[∘]=

$\{\{x \to 0\}, \{x \to 1\}, \{x \to -\sqrt{3}\}, \{x \to \sqrt{3}\}\}$

Out[∘]=

$(-3 + x)\, (-1 + x)\, x\, (1 + x)$

Out[∘]=

$\{\{x \to -1\}, \{x \to 0\}, \{x \to 1\}, \{x \to 3\}\}$

We will now find the corresponding masses. First we compute the Vandermonde matrices and their inverses.

In[∘]:=
```
V1 =
  Table[root^i, {i, 0, 3}, {root, x /. Solve[Det[polH1[x]] == 0, x]}] /. Indeterminate → 1;
MatrixForm[V1]
InvV1 = Inverse[V1]; MatrixForm[InvV1]
V2 =
  Table[root^i, {i, 0, 3}, {root, x /. Solve[Det[polH2[x]] == 0, x]}] /. Indeterminate → 1;
MatrixForm[V2]
InvV2 = Inverse[V2];
MatrixForm[InvV2]
```

Power: Indeterminate expression 0^0 encountered.

Out[∘]//MatrixForm=

$$\begin{pmatrix} 1 & 1 & 1 & 1 \\ 0 & 1 & -\sqrt{3} & \sqrt{3} \\ 0 & 1 & 3 & 3 \\ 0 & 1 & -3\sqrt{3} & 3\sqrt{3} \end{pmatrix}$$

Out[∘]//MatrixForm=

$$\begin{pmatrix} 1 & -1 & -\frac{1}{3} & \frac{1}{3} \\ 0 & \frac{3}{2} & 0 & -\frac{1}{2} \\ 0 & \frac{3-3\sqrt{3}}{12\sqrt{3}} & \frac{1}{6} & \frac{-3+\sqrt{3}}{12\sqrt{3}} \\ 0 & \frac{-3-3\sqrt{3}}{12\sqrt{3}} & \frac{1}{6} & \frac{3+\sqrt{3}}{12\sqrt{3}} \end{pmatrix}$$

Power: Indeterminate expression 0^0 encountered.

Out[∘]//MatrixForm=

$$\begin{pmatrix} 1 & 1 & 1 & 1 \\ -1 & 0 & 1 & 3 \\ 1 & 0 & 1 & 9 \\ -1 & 0 & 1 & 27 \end{pmatrix}$$

Out[∘]//MatrixForm=

$$\begin{pmatrix} 0 & -\frac{3}{8} & \frac{1}{2} & -\frac{1}{8} \\ 1 & -\frac{1}{3} & -1 & \frac{1}{3} \\ 0 & \frac{3}{4} & \frac{1}{2} & -\frac{1}{4} \\ 0 & -\frac{1}{24} & 0 & \frac{1}{24} \end{pmatrix}$$

We compute the corresponding masses by InvV1 {S0,S1,S2,Z1} and InvV2 {S0,S1,S2,Z2}.

In[∘]:=
```
masses1 = Table[InvV1[[r]].{S0, S1, S2, Z1}, {r, 1, Length[InvV1]}];
MatrixForm[masses1]
masses1Simplified = FullSimplify[ArrayFlatten[Transpose[{masses1}]]];
MatrixForm[masses1Simplified]
masses2 = Table[InvV2[[r]].{S0, S1, S2, Z2}, {r, 1, Length[InvV2]}];
MatrixForm[masses2]
masses2Simplified = FullSimplify[ArrayFlatten[Transpose[{masses2}]]];
MatrixForm[masses2Simplified]
```

Out[∘]//MatrixForm=

$$\begin{pmatrix} \begin{pmatrix} 0 \\ 0 \end{pmatrix} & \begin{pmatrix} 0 \\ 1 \end{pmatrix} \\ \begin{pmatrix} 2 \\ 2 \end{pmatrix} & \begin{pmatrix} 2 \\ 2 \end{pmatrix} \\ \begin{pmatrix} \frac{25}{3} + \frac{3-3\sqrt{3}}{6\sqrt{3}} + \frac{-3+\sqrt{3}}{6\sqrt{3}} \\ \frac{13}{3} + \frac{3-3\sqrt{3}}{6\sqrt{3}} + \frac{-3+\sqrt{3}}{6\sqrt{3}} \end{pmatrix} & \begin{pmatrix} \frac{13}{3} + \frac{3-3\sqrt{3}}{6\sqrt{3}} + \frac{-3+\sqrt{3}}{6\sqrt{3}} \\ \frac{7}{3} + \frac{3-3\sqrt{3}}{6\sqrt{3}} + \frac{-3+\sqrt{3}}{6\sqrt{3}} \end{pmatrix} \\ \begin{pmatrix} \frac{25}{3} + \frac{-3-3\sqrt{3}}{6\sqrt{3}} + \frac{3+\sqrt{3}}{6\sqrt{3}} \\ \frac{13}{3} + \frac{-3-3\sqrt{3}}{6\sqrt{3}} + \frac{3+\sqrt{3}}{6\sqrt{3}} \end{pmatrix} & \begin{pmatrix} \frac{13}{3} + \frac{-3-3\sqrt{3}}{6\sqrt{3}} + \frac{3+\sqrt{3}}{6\sqrt{3}} \\ \frac{7}{3} + \frac{-3-3\sqrt{3}}{6\sqrt{3}} + \frac{3+\sqrt{3}}{6\sqrt{3}} \end{pmatrix} \end{pmatrix}$$

Out[∘]//MatrixForm=

$$\begin{pmatrix} 0 & 0 \\ 0 & 1 \\ 2 & 2 \\ 2 & 2 \\ 8 & 4 \\ 4 & 2 \\ 8 & 4 \\ 4 & 2 \end{pmatrix}$$

Out[∘]//MatrixForm=

$$\begin{pmatrix} \begin{pmatrix} 12 \\ 6 \end{pmatrix} & \begin{pmatrix} 6 \\ 3 \end{pmatrix} \\ \begin{pmatrix} 0 \\ 0 \end{pmatrix} & \begin{pmatrix} 0 \\ 1 \end{pmatrix} \\ \begin{pmatrix} 2 \\ 2 \end{pmatrix} & \begin{pmatrix} 2 \\ 2 \end{pmatrix} \\ \begin{pmatrix} 4 \\ 2 \end{pmatrix} & \begin{pmatrix} 2 \\ 1 \end{pmatrix} \end{pmatrix}$$

Out[∘]//MatrixForm=

$$\begin{pmatrix} 12 & 6 \\ 6 & 3 \\ 0 & 0 \\ 0 & 1 \\ 2 & 2 \\ 2 & 2 \\ 4 & 2 \\ 2 & 1 \end{pmatrix}$$

In[∘]:=

In[∘]:=

In[ ]:=

In[ ]:=