
Computations supporting Example 3.6 in the paper Matricial Gaussian Quadrature Rules: Nonsingular Case by Aljaž Zalar and Igor Zbovič.

This example shows studies a linear operator L which is represented by a measure μ which contains the atom $t = 0$ with multiplicity 1. However, since $\text{rank}(\text{HH} \ K) = \text{rank}(\text{HH})$, it follows that L is also represented by a measure $\tilde{\mu}$ which contains the atom $t = 0$ atom with multiplicity $p = 2$.

Let $\mu = \sum_{j=1}^4 x_j A_j$ be a finitely atomic matrix measure with $(x_1, x_2, x_3, x_4) = (0, 1, -1, 2)$ and $(A_1, A_2, A_3, A_4) = \left(\begin{pmatrix} 2 & 2 \\ 2 & 2 \end{pmatrix}, \begin{pmatrix} 1 & 1 \\ 1 & 1 \end{pmatrix}, \begin{pmatrix} 0 & 0 \\ 0 & 1 \end{pmatrix}, \begin{pmatrix} 1 & 0 \\ 0 & 0 \end{pmatrix} \right)$, and let L be a linear operator, defined by $L(f) = \int f d\mu$.

```
In[ ]:= x1 = 0
        x2 = 1
        x3 = -1
        x4 = -2
        atoms = {x1, x2, x3, x4};
        A1 = {{2, 2}, {2, 2}};
        MatrixForm[A1]
        A2 = {{1, 1}, {1, 1}};
        MatrixForm[A2]
        A3 = {{0, 0}, {0, 1}};
        MatrixForm[A3]
        A4 = {{1, 0}, {0, 0}};
        MatrixForm[A4]
        masses = {A1, A2, A3, A4};
```

```
Out[ ]:=
```

0

```
Out[ ]:=
```

1

```
Out[ ]:=
```

-1

```
Out[ ]:=
```

-2

```
Out[ ]//MatrixForm=
```

$$\begin{pmatrix} 2 & 2 \\ 2 & 2 \end{pmatrix}$$

```
Out[ ]//MatrixForm=
```

$$\begin{pmatrix} 1 & 1 \\ 1 & 1 \end{pmatrix}$$

```
Out[ ]//MatrixForm=
```

$$\begin{pmatrix} 0 & 0 \\ 0 & 1 \end{pmatrix}$$

```
Out[ ]//MatrixForm=
```

$$\begin{pmatrix} 1 & 0 \\ 0 & 0 \end{pmatrix}$$

For every $i \in \{0, 1, 2\}$ we define $S_i := L(x^i)$.

```
In[ ]:= f0[x_] := 1;
        f1[x_] := x;
        f2[x_] := x^2;
        S0 = Total[masses * (f0 /@ atoms)];
        MatrixForm[S0]
        S1 = Total[masses * (f1 /@ atoms)];
        MatrixForm[S1]
        S2 = Total[masses * (f2 /@ atoms)];
        MatrixForm[S2]
```

```
Out[ ]//MatrixForm=
```

$$\begin{pmatrix} 4 & 3 \\ 3 & 4 \end{pmatrix}$$

```
Out[ ]//MatrixForm=
```

$$\begin{pmatrix} -1 & 1 \\ 1 & 0 \end{pmatrix}$$

```
Out[ ]//MatrixForm=
```

$$\begin{pmatrix} 5 & 1 \\ 1 & 2 \end{pmatrix}$$

We define the moment matrix $M := M(1)$, the $(x - t)$ -localizing matrix HH , and the matrix K .

```
In[ ]:= M = ArrayFlatten[{{S0, S1}, {S1, S2}}];
        MatrixForm[M]
        HH = S1;
        MatrixForm[HH]
        K = S2;
        MatrixForm[K]
```

```
Out[ ]//MatrixForm=
```

$$\begin{pmatrix} 4 & 3 & -1 & 1 \\ 3 & 4 & 1 & 0 \\ -1 & 1 & 5 & 1 \\ 1 & 0 & 1 & 2 \end{pmatrix}$$

```
Out[ ]//MatrixForm=
```

$$\begin{pmatrix} -1 & 1 \\ 1 & 0 \end{pmatrix}$$

```
Out[ ]//MatrixForm=
```

$$\begin{pmatrix} 5 & 1 \\ 1 & 2 \end{pmatrix}$$

We check that M is positive definite by computing its eigenvalues.

```
In[ ]:= N[Eigenvalues[M]]
Out[ ]:=
```

$$\{7.10792, 5.59462, 2.27536, 0.0221038\}$$

Although the measure contains the atom $t = 0$, its multiplicity is $\text{rank}(A_1) = 1$.

However, the localizing matrix HH is invertible, therefore $\text{rank}(HH - K) = \text{rank}(HH)$, and thus there exists a 4-atomic \mathbb{R} -representing measure $\tilde{\mu}$ for L which contains the atom $t = 0$ with multiplicity 2. Such measure $\tilde{\mu}$ is unique and we will now find its atoms. We first compute $Z := S_3^{(\tilde{\mu})} = K^T HH^{-1}K$.

```
In[ ]:= Z = Transpose[K].Inverse[HH].K;
MatrixForm[Z]
Out[ ]//MatrixForm=

$$\begin{pmatrix} 11 & 13 \\ 13 & 8 \end{pmatrix}$$

```

We now obtain the generating polynomial $\text{polH}(x)$.

```
In[ ]:= H = Inverse[M].ArrayFlatten[{{S2}, {Z}}];
MatrixForm[H]
H0 = H[[1 ;; 2]];
MatrixForm[H0]
H1 = H[[3 ;; 4]];
MatrixForm[H1]
polH[x_] := x^2 IdentityMatrix[2] - (H0 + x H1)
polH[x_] // Simplify // MatrixForm
Out[ ]//MatrixForm=

$$\begin{pmatrix} 0 & 0 \\ 0 & 0 \\ 1 & 2 \\ 6 & 3 \end{pmatrix}$$

Out[ ]//MatrixForm=

$$\begin{pmatrix} 0 & 0 \\ 0 & 0 \end{pmatrix}$$

Out[ ]//MatrixForm=

$$\begin{pmatrix} 1 & 2 \\ 6 & 3 \end{pmatrix}$$

Out[ ]//MatrixForm=

$$\begin{pmatrix} (-1 + x_-) x_- & -2 x_- \\ -6 x_- & (-3 + x_-) x_- \end{pmatrix}$$

```

We compute the determinant from which we get the atoms.

```
In[ ]:= Factor[Det[polH[x]]]
Solve[Det[polH[x]] == 0, x]
Out[ ]:=

$$x^2 (-9 - 4x + x^2)$$

Out[ ]:=

$$\{\{x \rightarrow 0\}, \{x \rightarrow 0\}, \{x \rightarrow 2 - \sqrt{13}\}, \{x \rightarrow 2 + \sqrt{13}\}\}$$

```

We will now find the corresponding masses. First we compute the Vandermonde matrix and its inverse.

```
In[*]:= newAtoms = DeleteDuplicates[x /. Solve[Det[polH[x]] == 0, x]]
V = Table[root^i, {i, 0, 2}, {root, newAtoms}] /. Indeterminate -> 1;
MatrixForm[V]
InvV = Inverse[V];
MatrixForm[InvV]
```

```
Out[*]:= {0, 2 - √13, 2 + √13}
```

 **Power:** Indeterminate expression 0^0 encountered. 

```
Out[*]//MatrixForm=
```

$$\begin{pmatrix} 1 & 1 & 1 \\ 0 & 2 - \sqrt{13} & 2 + \sqrt{13} \\ 0 & (2 - \sqrt{13})^2 & (2 + \sqrt{13})^2 \end{pmatrix}$$

```
Out[*]//MatrixForm=
```

$$\begin{pmatrix} 1 & \frac{4}{9} & -\frac{1}{9} \\ 0 & -\frac{17+4\sqrt{13}}{18\sqrt{13}} & -\frac{-2-\sqrt{13}}{18\sqrt{13}} \\ 0 & -\frac{-17+4\sqrt{13}}{18\sqrt{13}} & -\frac{2-\sqrt{13}}{18\sqrt{13}} \end{pmatrix}$$

We compute the corresponding masses by $\text{InvV} \{S_0, S_1, S_2\}$.

```
In[*]:= newMasses = Table[InvV[[r]].{S0, S1, S2}, {r, 1, Length[InvV]}];
newMassesSimplified = FullSimplify[ArrayFlatten[Transpose[{newMasses}]]];
rationalizeExpr[expr_] := PowerExpand[Together[Rationalize[expr, 0]]];
newMassesRationalized = Map[rationalizeExpr, newMassesSimplified, {2}];
MatrixForm[newMassesRationalized]
```

```
Out[*]//MatrixForm=
```

$$\begin{pmatrix} 3 & \frac{10}{3} \\ \frac{10}{3} & \frac{34}{9} \\ \frac{1}{26} (13 + 3\sqrt{13}) & \frac{1}{78} (-13 - 5\sqrt{13}) \\ \frac{1}{78} (-13 - 5\sqrt{13}) & \frac{1}{117} (13 + 2\sqrt{13}) \\ \frac{1}{26} (13 - 3\sqrt{13}) & \frac{1}{78} (-13 + 5\sqrt{13}) \\ \frac{1}{78} (-13 + 5\sqrt{13}) & \frac{1}{117} (13 - 2\sqrt{13}) \end{pmatrix}$$